
ABSTRACT  
PROC REPORT with the combination of ODS RTF is a powerful tool and widely used to produce tables and listings 
for clinical trial data in pharmaceutical industry. Limitation of ODS RTF output is that the content of same group 
variables are not carried forward to following page for SAS® Versions 8.2 and 9.1. In this paper, a simple macro 
based solution will be presented to help SAS® users to overcome the limitation of ODS RTF. 

INTRODUCTION  
Since ODS RTF has been available in production in SAS® Version 8.1, it has become the most useful function to 
programmers in biotechnology/pharmaceutical industries.  In this paper we will look at PROC REPORT output to the 
RTF destination, as designed to overcome an implication that programmers need to anticipate when creating tables 
and listings that extend over more than one page. For SAS® Versions 8.2 and 9.1, PROC REPORT output will not 
repeat GROUP/ORDER variable values at the start of a new page. In this paper I will outline the approach that I have 
adopted to overcome the limitation of the ODS RTF in my day-to-day work as a statistical programmer. 

SOLUTION 
The proposed macro based solution is simple and no advanced SAS® programming skill is needed.  The basic idea 
of this macro is to control number of observations that fit to each page by taking advantage of automatic variable _n_.   
 
An example program presented in this paper is to produce a summary table of lab parameters for each treatment 
group. The n, mean, std, median, min, max are presented for each lab parameter and visit, and the table output has 
treatment groups listed top of the table and  lab parameter, visit and summary statistics are  listed left side of the 
table.  
 
Step 1 
The first step is to sort the input data of PROC REPORT by the group/order variables.   
 
proc sort data=lab; 
 by test visit ;  
run;  
 
Step 2 
The second step is to set up a macro to create a variable upon which a page break is forced in PROC REPORT. The 
input parameters for this macro are as follows: 
 

PARAMETER  DESCRIPTION  
inds  the input dataset name  
byvars  by variables that are sorted from 

step 1  
brkpt  a variable upon which a page 

break is forced 
 
Step 3 
The macro completes following steps in the process. 
1) assign the group/order variables that are sorted  
2) creates a variable pagebrk based on number of observations of brkpt 
3) brkpt value of 18 is picked in order to increment pagebrk after every 18 observations of input dataset 
 
%macro pagebreak (inds=, byvars=, brkpt=); 
     /* one page can hold 18 lines of output in this scenario */                      
    data &inds(drop=brk);  
     set &inds; 
     by &byvars; 
     if _n_ eq 1 then brk=0; 
     else brk+1; 
     /* when brk is greater than or equal to 18, it gets reset to 0 and increments 
pagebrk by 1 */ 
          if brk >=&brkpt then do;  
     pagebrk+1; brk=0; 
     end; 



run; 
%mend; 
%pagebreak (inds=lab, byvars=test visit, brkpt=18) 
 
Step 4 
The output data set now contains all the information required to produce an accurately paginated table to retain the 
group variable values to following page using the PROC REPORT procedure: 
 
ods listing close;         
ods rtf file="C:\ODS RTF\lab_summary.rtf" style=rtf_style;  
title1 " ";  
title2 "Summary of Laboratory Values ";  
options nodate nonumber;  
proc report data=lab nowindows headline headskip split='^' 
style(header)=[background=transp]; 
column pagebrk TEST VISIT STAT varA varD varE; 
/* pagebrk is set to noprint as it will be used for only pagination */ 
define pagebrk /order noprint; 
define TEST /group "Test"  style(column)=[just=L asis=on cellwidth=30%] 
style(header)=[Just=L]; 
define VISIT /group "Visit"   order=internal style(column)=[just=c asis=on 
cellwidth=12%] style(header)=[Just=c]; 
define STAT /  "Statistics" style(column)=[just=L asis=on cellwidth=8%] 
style(header)=[Just=L ]; 
define varA  / "Group A" style(column)=[just=L asis=on cellwidth=10%] 
style(header)=[Just=c]; 
define varD  / "Group D" style(column)=[just=L asis=on cellwidth=10%] 
style(header)=[Just=c]; 
define varE  / "Group E" style(column)=[just=L asis=on cellwidth=10%] 
style(header)=[Just=c]; 
break after pagebrk/page; 
run; 
ods rtf close; 
ods listing; 
 
Use of the pagebrk variable as the left most in COLUMN statement, invisible order variable in DEFINE statement, and 
the BREAK AFTER statement are the only additional lines required to achieve the accurate pagination of the 
summary report. 

LIMITATIONS  
One of the tasks involved in this technique is the permutation and combination of number of observations of output in 
each page. This method is straight-forward to set page breaks when each observation only appears on one line. 
However, in some tables or listings the length of some data values causes text wrapping within the cells. In this case, 
SAS® does not know how many observations will fit on each page. To overcome this problem, in version 9.2 of 
SAS®, you can try the new option of SPANROWS in the PROC REPORT statement along with trying the new 
tagsets.rtf as well. 

CONCLUSION 
Using the automatic variable _n_ can be a great way to control number of observations that fit to each page using 
ODS RTF and PROC REPORT. However, there are some problems that you need to be aware of in creating tables or 
listings with this method. This paper demonstrated to simplify your coding and effectively overcome some of your 
practical problems.   

